

Simple Rectangle-based Functional Programs for Computing Reflexive-transitive Closures

Rudolf Berghammer

Institut für Informatik
Christian-Albrechts-Universität zu Kiel

Joint work with Sebastian Fischer

Reachability Closures of Relation

Given a relation R on a set X ,

- its *reflexive-transitive closure* is $R^* = \bigcup_{n \geq 0} R^n$,
- its *transitive closure* is $R^+ = \bigcup_{n \geq 1} R^n$.

R^* specifies **reachability** in the graph $g = (X, R)$ and R^+ specifies **reachability via non-empty paths**.

Fundamental properties:

- $O^* = I$
- $R^* = I \cup R^+$
- $(R \cup S)^* = R^*(SR^*)^*$ (**star-decomposition rule**)
- R is transitive iff $R = R^+$

Warshall's Algorithm

- Traditional imperative computation of R^* using a representation of relations by 2-dimensional Boolean arrays.

```
Q := R;  
for  $i \in X$  do  
  for  $j \in X$  do  
    for  $k \in X$  do  
       $Q[j, k] := Q[j, k] \vee (Q[j, i] \wedge Q[i, k])$   
for  $i \in X$  do  
   $Q[i, i] := \text{True}$ 
```

} Warshall

- Arrays are unfit for representing relations if they are of “medium density” or even sparse.

Computational linguistics, XML-query processing, order theory, ...

Here successor lists are much more economic; but such a representation sacrifices the simplicity and efficiency of the algorithm.

- Arrays with side-effects are also problematic if the functional paradigm is used.

Aim of the Talk

To show how systematically to derive simple functional programs for computing reflexive-transitive closures that

- base on a common **schematic algorithm**,
- use a representation of relations via **successor lists**,
- have for specific instantiations the same **cubic running time** as the imperative algorithm.

The used tools and techniques are as in the case of the RAMiCS 12 talk:

- **Relation algebra** for problem specification, the derivation of the generic algorithm and its specializations.
- **Data refinement** to obtain list representations and for the translation into Haskell.

For visualization purposes we depict relations as **Boolean matrices**, drawn by the computer system **RELVIEW**.

Relation Algebra

Relations:


- If R is a relation with source X and target Y , we write $R : X \leftrightarrow Y$.
- $[X \leftrightarrow Y]$ is the *type/set* of all relations with source X and target Y .

Signature of relation algebra:

- Constants: $0, L, I$.
- Operations: $R \cup S, R \cap S, RS, \overline{R}, R^T$.
- Tests: $R \subseteq S, R = S$.

Properties;

- *Reflexivity*: $I \subseteq R$.
- *Transitivity*: $RR \subseteq R$.
- *Vector*: $vL = v$.
- *Point*: $pL = p, Lp = L$ (*surjectivity*) and $pp^T \subseteq I$ (*injectivity*).




Rectangles and Reachability Closures


- A relation $S : X \leftrightarrow X$ is a *rectangle* if there exist $A, B \subseteq X$ such that

$$S = A \times B.$$


- Examples, where $X = \{1, \dots, 8\}$:



R



S_1



S_2

R is **not a rectangle**. S_1 is **a rectangle** since $S_1 = \{1\} \times \{2\}$, and S_2 is **a rectangle** since $S_2 = \{3, 4, 6\} \times \{1, 3, 5, 7\}$.

- Relation-algebraic specification: $S : X \leftrightarrow X$ is a rectangle iff

$$SLS \subseteq S.$$

Theorem. Assume $R : X \leftrightarrow X$ and let $S : X \leftrightarrow X$ be a **rectangle**. Then

$$(R \cup S)^* = R^* \cup R^* S R^*.$$

Proof. Transitivity $SR^*SR^* \subseteq SRSR^* \subseteq SR^*$ yields $(SR^*)^+ = SR^*$. By means of this equation, the statement follows from

$$\begin{aligned} (R \cup S)^* &= R^* (SR^*)^* && \text{star-decomposition} \\ &= R^* (I \cup (SR^*)^+) \\ &= R^* (I \cup SR^*) && SR^* \text{ transitive} \\ &= R^* \cup R^* S R^* && \text{distributivity.} \end{aligned}$$

Generic Algorithm. The following functional algorithm computes R^* :

```

rtc : [X ↔ X] → [X ↔ X]
rtc(R) = if R = 0 then I
           else let S = rectangle(R)
                C = rtc(R ∩ S̄)
           in C ∪ CSC
    
```

Here $rectangle(R)$ yields a non-empty rectangle S with $S \subseteq R$.

Possibilities for $rectangle(R)$:

- Selection of an atomic relation contained in R .
- Selection of a relation that singles out a row of R .
- Selection of a relation that singles out a column of R .
- Selection of a maximal (a non-enlargable) rectangle contained in R
 - started vertically
 - started horizontally.

Examples, where $X = \{1, \dots, 8\}$:

	1	2	3	4	5	6	7	8
1		■						■
2								
3	■		■		■			
4	■		■		■			
5								
6	■		■		■			
7								
8			■					■

R

	1	2	3	4	5	6	7	8
1		■						
2								
3								
4								
5								
6								
7								
8								

S_1

	1	2	3	4	5	6	7	8
1								
2								
3								
4	■		■		■		■	
5								
6								
7								
8								

S_2

	1	2	3	4	5	6	7	8
1								
2								
3			■					
4			■					
5								
6			■					
7								
8			■					

S_3

	1	2	3	4	5	6	7	8
1								
2								
3	■		■		■		■	
4	■		■		■		■	
5								
6	■		■		■		■	
7								
8								

S_4


Singling Out Rows

For $R : X \leftrightarrow X$ and a *point* $p : X \leftrightarrow \mathbf{1}$ with $p \subseteq RL$ (“it has successors”), we define:

$$S := pp^T R$$

The rectangle $S = pp^T R$ corresponds to the row of R designated by the point p and $R \cap \bar{S}$ “zeroes out” out this row.


Example, where $X = \{1, \dots, 8\}$ and p describes the element 4 of X :




R



p



$S = pp^T R$



$R \cap \bar{S}$

Proof of the rectangle property of S (using the *vector property* of p):

$$SLS = pp^T RLpp^T R \subseteq pLp^T R = pp^T R = S$$

Proof of the inclusion $S \subseteq R$ (using the *injectivity* of p):

$$S = pp^T R \subseteq IR = R$$

Proof of non-emptiness of S by contradiction (using the *surjectivity* of p):

$$S = O \iff pp^T R \subseteq O \iff \dots \iff p \subseteq \overline{RL}$$

Refined Algorithm. Singling out a row leads to:

```
rtc : [X ↔ X] → [X ↔ X]
rtc(R) = if R = O then I
           else let p = point(RL)
                S = ppTR
                C = rtc(R ∩  $\overline{S}$ )
           in C ∪ CSC
```

Here $point(RL)$ yields a point p with $p \subseteq RL$.

From Relation Algebra to Functions

Represent $F : X \leftrightarrow X$ by $f : X \rightarrow 2^X$ such that $(x, y) \in F$ iff $y \in f(x)$.

- The empty relation O is represented by

$$\lambda x. \emptyset.$$

- The identity relation I is represented by

$$\lambda x. \{x\}.$$

- If $R \neq O$ is represented by r , then the choice of a point p with $p \subseteq RL$ corresponds to the choice of an element $n \in X$ with $r(n) \neq \emptyset$.

- With r and n from above, $R \cap \overline{pp^T R}$ is represented by

$$r[n \leftarrow \emptyset].$$

- If C is represented by c , then $C \cup Cpp^T RC$ is represented by

$$\lambda x. \mathbf{if} \ n \in c(x) \ \mathbf{then} \ c(x) \cup \bigcup \{c(k) \mid k \in r(n)\} \ \mathbf{else} \ c(x).$$

Let $ks := \bigcup \{c(k) \mid k \in r(n)\}$. Then the last representation follows from

$$(x, y) \in C \cup Cpp^T RC$$

$$\Leftrightarrow (x, y) \in C \vee (x, y) \in Cpp^T RC$$

$$\Leftrightarrow (x, y) \in C \vee \exists i : (x, i) \in C \wedge \exists j : (i, j) \in pp^T \wedge (j, y) \in RC$$

$$\Leftrightarrow (x, y) \in C \vee \exists i : (x, i) \in C \wedge \exists j : i = n \wedge j = n \wedge (j, y) \in RC$$

$$\Leftrightarrow (x, y) \in C \vee ((x, n) \in C \wedge (n, y) \in RC)$$

$$\Leftrightarrow (x, y) \in C \vee ((x, n) \in C \wedge \exists k : (n.k) \in R \wedge (k, y) \in C)$$

$$\Leftrightarrow y \in c(x) \vee \mathbf{if} \ n \in c(x) \ \mathbf{then} \ \exists k : k \in r(n) \wedge y \in c(k) \ \mathbf{else} \ \mathit{false}$$

$$\Leftrightarrow y \in c(x) \vee \mathbf{if} \ n \in c(x) \ \mathbf{then} \ y \in \bigcup \{c(k) \mid k \in r(n)\} \ \mathbf{else} \ y \in \emptyset$$

$$\Leftrightarrow y \in c(x) \vee y \in \mathbf{if} \ n \in c(x) \ \mathbf{then} \ y \in ks \ \mathbf{else} \ \emptyset$$

$$\Leftrightarrow y \in \mathbf{if} \ n \in c(x) \ \mathbf{then} \ c(x) \cup ks \ \mathbf{else} \ c(x)$$

$$\Leftrightarrow y \in (\lambda x. \mathbf{if} \ n \in c(x) \ \mathbf{then} \ c(x) \cup ks \ \mathbf{else} \ c(x))(x)$$

using that $(i, j) \in pp^T$ iff $i = n$ and $j = n$.

1	
2	
3	
4	
5	
6	
7	
8	

1							
2							
3							
4							
5							
6							
7							
8							

Refined Algorithm on Functions.

We replace

- all relations by their representing functions,
- the choice of p by that of n .

Doing so, we arrive at the following **refined algorithm on functions**:

$$\begin{aligned} rtc &: (X \rightarrow 2^X) \rightarrow (X \rightarrow 2^X) \\ rtc(r) &= \mathbf{if} \ r = \lambda x. \emptyset \ \mathbf{then} \ \lambda x. \{x\} \\ &\quad \mathbf{else} \ \mathbf{let} \ n = elem(\{x \in X \mid r(x) \neq \emptyset\}) \\ &\quad \quad c = rtc(r[n \leftarrow \emptyset]) \\ &\quad \quad ks = \bigcup \{c(k) \mid k \in r(n)\} \\ &\quad \mathbf{in} \ \lambda x. \mathbf{if} \ n \in c(x) \ \mathbf{then} \ c(x) \cup ks \ \mathbf{else} \ c(x) \end{aligned}$$

Here $elem(\{x \in X \mid r(x) \neq \emptyset\})$ yields a $n \in X$ with $r(n) \neq \emptyset$.

List Representation of Output Functions

Assuming $X = \{0, \dots, m\}$ we represent $f : X \rightarrow 2^X$ by $[f(0), \dots, f(m)]$.

- For $\lambda x.\{x\}$ we get the representation

$$[\{x\} \mid x \in [0..m]].$$

- For $\lambda x.\mathbf{if} \ n \in c(x) \ \mathbf{then} \ c(x) \cup ks \ \mathbf{else} \ c(x)$ we get the representation

$$[\mathbf{if} \ n \in ms \ \mathbf{then} \ ms \cup ks \ \mathbf{else} \ ms \mid ms \in cs],$$

where $cs \in (2^X)^*$ represents c . Note that then $c(k)$ becomes $cs!!k$.

Moving from the output functions to the list representations, we get:

$$rtc : (X \rightarrow 2^X) \rightarrow (2^X)^*$$

$$rtc(r) = \mathbf{if} \ r = \lambda x.\emptyset \ \mathbf{then} \ [\{x\} \mid x \in [0..m]]$$

$$\mathbf{else} \ \mathbf{let} \ n = \mathit{elem}(\{x \in X \mid r(x) \neq \emptyset\})$$

$$cs = rtc(r[n \leftarrow \emptyset])$$

$$ks = \bigcup \{cs!!k \mid k \in r(n)\}$$

$$\mathbf{in} \ [\mathbf{if} \ n \in ms \ \mathbf{then} \ ms \cup ks \ \mathbf{else} \ ms \mid ms \in cs]$$

List Representation of Input Functions

We represent $f : X \rightarrow 2^X$ as list of the pairs $(x, f(x))$ for which $f(x) \neq \emptyset$.

- If $rs \in (X \times 2^X)^*$ represents r , testing $r = \lambda x. \emptyset$ reduces to

$$rs = [].$$

- An n with $r(n) \neq \emptyset$ is then given by the first component of $head(rs)$.
- The list representation of $r[n \leftarrow \emptyset]$ is then

$$tail(rs).$$

If we use pattern matching in the **let**-clause, we get:

$$rtc : (X \times 2^X)^* \rightarrow (2^X)^*$$

$$rtc(rs) = \mathbf{if} \ rs = [] \ \mathbf{then} \ [\{x\} \mid x \in [0..m]] \\ \mathbf{else} \ \mathbf{let} \ (n, ns) = head(rs) \\ \quad cs = rtc(tail(rs)) \\ \quad ks = \bigcup \{cs!!k \mid k \in ns\} \\ \mathbf{in} \ [\mathbf{if} \ n \in ms \ \mathbf{then} \ ms \cup ks \ \mathbf{else} \ ms \mid ms \in cs]$$

Refined Algorithm on Lists of Sets

The last version of *rtc* also can be written in the following form.

- The auxiliary function *step* performs the essential computations of the recursion.

$$\begin{aligned} \textit{step} &: (X \times 2^X) \times (2^X)^* \rightarrow (2^X)^* \\ \textit{step}((n, ns), cs) &= \mathbf{let} \ ks = \bigcup \{cs!!k \mid k \in ns\} \\ &\quad \mathbf{in} \ [\mathbf{if} \ n \in ms \ \mathbf{then} \ ms \cup ks \ \mathbf{else} \ ms \mid ms \in cs] \end{aligned}$$

- The main program:

$$\begin{aligned} \textit{rtc} &: (X \times 2^X)^* \rightarrow (2^X)^* \\ \textit{rtc}(rs) &= \mathbf{if} \ rs = [] \ \mathbf{then} \ [\{x\} \mid x \in [0..m]] \\ &\quad \mathbf{else} \ \textit{step}(\textit{head}(rs), \textit{rtc}(\textit{tail}(rs))) \end{aligned}$$

Translation into Haskell

Let a Haskell constant `m` of type `Int` for m be at hand.

An obvious implementation of subsets of X in Haskell is given by *sorted lists* over X *without multiple occurrences of elements*.

- \emptyset is implemented by `[]`.
- Set-membership is implemented by `elem`.
- An implementation of set union using linear running time is

```
merge :: [Int] -> [Int] -> [Int]
merge [] ys = ys
merge xs [] = xs
merge (x:xs) (y:ys) =
    case compare x y of EQ -> x : merge xs ys
                       LT -> x : merge xs (y:ys)
                       GT -> y : merge (x:xs) ys
```

(merging and removal of multiple occurrences of elements).

The Haskell list for $\bigcup\{cs!!k \mid k \in ns\}$ is obtained by merging all sorted lists $cs!!k$, where k ranges over the elements of ns . Haskell code for *step*:

```
step :: (Int, [Int]) -> [[Int]] -> [[Int]]
step (n, ns) cs =
  let ks = foldr merge [] [cs!!k | k <- ns]
  in  [if elem n ms then merge ms ks else ms | ms <- cs]
```

The running time is $O(m^2)$ since during the whole merging process each argument and result of merge has at most length $m + 1$.

Also the main function *rtc* can be seen as a right-fold over the Haskell counterpart *rs* of *rs*.

```
rtc :: [(Int, [Int])] -> [[Int]]
rtc rs =
  foldr step [[x] | x <- [0..m]] rs
```

Slightly dissatisfying is that the Haskell function `rtc`

- depends on the constant `m`
- works with two different list representations of relations.

Because of

$$\text{step } (n, []) \text{ cs} = \text{cs}$$

we can change it in such a way that it is independent of `m` and also the input `rs` is a list of type `[[Int]]` that contains as x -th component the (possible empty) successor list of x .

```
rtc :: [[Int]] -> [[Int]]
rtc rs =
  let xs = [0..length rs - 1]
  in foldr step [[n] | n <- xs] (zip xs rs)
```

The complexity of the entire program is $O(m^3)$.

The Complete Haskell Program

```
merge :: [Int] -> [Int] -> [Int]
merge [] ys = ys
merge xs [] = xs
merge (x:xs) (y:ys) =
    case compare x y of EQ -> x : merge xs ys
                       LT -> x : merge xs (y:ys)
                       GT -> y : merge (x:xs) ys

step :: (Int, [Int]) -> [[Int]] -> [[Int]]
step (n, ns) cs =
    let ks = foldr merge [] [cs!!k | k <- ns]
    in [if elem n ms then merge ms ks else ms | ms <- cs]

rtc :: [[Int]] -> [[Int]]
rtc rs =
    let xs = [0..length rs - 1]
    in foldr step [[n] | n <- xs] (zip xs rs)
```

Concluding Remarks

Some **improvements with regard to practical running times** are possible.

- A user-defined `elem-test` that takes advantage of the fact that the successor lists are sorted.
- A linear time computation of `[cs!!k | k <- ns]` that also uses that all successor lists are sorted.
- Transformation into tail-recursive version.

Further improvement:

- Replace `zip xs rs` by

```
zip xs (map (nub.sort) rs).
```

This is an implementation that **does not rely on the successor lists to be strictly increasing** without sacrificing time complexity.

Present and future work within the Kiel group:

Combination of
relation algebra and techniques of functional programming
for the formal development of efficient functional programs
on relation-based discrete structures.